How to keep PlotLegend expressions on top of each others even when placed at Bottom?

Version 9.

When writing

Plot[{Sin[x], Cos[x]}, {x, -2 Pi, 2 Pi},
PlotLegends -> Placed[TraditionalForm /@ {Sin[x], Cos[x]}, Right]]


The result is

Now I'd like to put the legend below the plot, but at the same time keep them looking the same way. i.e. sin(x) on top and cos(x) below. I do not want them to flatten. But when I do

Plot[{Sin[x], Cos[x]}, {x, -2 Pi, 2 Pi},
PlotLegends -> Placed[TraditionalForm /@ {Sin[x], Cos[x]}, Bottom]]


The result is

I can't use Grid or such to format these myself in the way I like them, since if I do, then PlotLegend will not see them any more. i.e. I can't just write something like

Plot[{Sin[x], Cos[x]}, {x, -2 Pi, 2 Pi},
PlotLegends -> Placed[ Column[{Sin[x], Cos[x]}], Bottom]]


Any one knows of a trick to do this? I looked at options for PlotLegend but do not see anything right away.

-

 Plot[{Sin[x], Cos[x]}, {x, -2 Pi, 2 Pi},
PlotLegends -> Placed[LineLegend["Expressions", LegendLayout -> "Column"], Bottom]]


Update: the following (which does not explicitly use LineLegend in the input) also gives the same picture:

 Plot[{Sin[x], Cos[x]}, {x, -2 Pi, 2 Pi},
PlotLegends ->  Placed[TraditionalForm /@ {Sin[x], Cos[x]}, Bottom]] /.
(LegendLayout -> _) :> LegendLayout -> "Column"


If you are certain that the strings "Row" and "Column" appear only in connection with legend layout in your plot, you can use the simpler replacement rule "Row" -> "Column".

-